



# Referring to a Variable Range with OFFSET

BY BILL JELEN

I received an interesting gadget as a birthday gift—a combination flashlight, tape measure, and cordless drill. It's rare for me to simultaneously need to measure distance while I'm drilling in the dark, so I rarely use all of the features at once. Since the tool can do three different things very well, however, it's a popular choice in my toolbox. In Excel, the OFFSET formula is a similar tool. It can do three different things, but it's rare that you would find a need to take advantage of all three at once.

## OFFSET's Features

Figure 1 illustrates an example of an OFFSET formula that uses all three of its features. It begins with a single cell reference (in red), directs Excel to move down five rows (blue arrow) and left four columns (purple arrow), then expands the selection to include a range of 24 cells that's eight rows high and three columns wide (green box).

The starting reference can be a single cell or a range of cells. Looking at the formula in Figure 1, the starting reference is cell F1. From there, you can then move a certain number of rows and columns: The 5 tells Excel to move down from F1 by

five rows, arriving at F6, and the -4 tells Excel to move left four columns from F6, finishing at B6.

The next two arguments are optional. From cell B6, you can specify that you want to refer to a range that is a certain number of rows tall and a certain number of columns

wide. In this case, Figure 1 specifies that OFFSET should refer to a range that is eight rows by three columns (B6:D13).

Since this use of OFFSET refers to a range, you can use it anywhere in a formula that would normally contain a range, such as =SUM(OFFSET(...)) or =VLOOKUP(A1,OFFSET(...),2,FALSE).


You're probably wondering when you would ever need to move from one cell to another cell and then resize the range. I agree—it's rare to use all of these features at once, but OFFSET is a handy tool to have available. Here are three examples that use the different features of OFFSET to solve common problems.

## Move Right by N Months

Suppose you have a table of monthly rates running from B6:M6 and want a formula to grab the rate for the current month. The formula =MONTH(TODAY()) will return a 10 if you open the workbook during October. As shown in Figure 2, you could use =OFFSET(A6,0,MONTH(TODAY())) to grab the October monthly rate.

In this example, you're specifying

Figure 1





**Figure 2**

=OFFSET(A6,0,MONTH(A1))													
	A	B	C	D	E	F	G	H	I	J	K	L	M
1	10/15/2008												
2	74												
3													
4	Rates by Month												
5		J	F	M	A	M	J	J	A	S	O	N	D
6		49	49	79	79	69	99	189	219	99		49	79
7													

**Figure 3**

=SMALL(OFFSET(A1,0,0,COUNT(A:A),1),2)				
	A	B	C	D
1	74		10	
2	29			
3	31			
4	65			
5	10			
6	23			
7	20			
8	9			
9	77			
10	27			
11	63			
12	36			
13				

a starting point and a column offset. You don't need to move to a different row, so the second parameter is 0. And since you only need to grab one cell, the optional fourth and fifth parameters are omitted.

**Have a Reference Expand as You Add More Data**

Perhaps you want to grab the second-smallest value from all the data enter-

ed in column A. If your data is numeric and has no blanks, the formula =OFFSET(A1,0,0,COUNT(A:A),1) will count how many numbers are in column A and expand the range to match the number of entries.

This formula returns a range, so you will usually enter that range as an argument in another function. In Figure 3, =SMALL(OFFSET(A1,0,0,COUNT(A:A),1),2) finds the second-smallest value entered in column A.

If the data includes numbers and text, use COUNTA instead of COUNT. If there potentially are blanks in a numeric column A, use MATCH(999999,A:A,1) to find the row number of the last cell in column A with a number. If there potentially are blanks in a text column A, use MATCH("ZZZZZ",A:A,1).

**Figure 4**

**Use OFFSET to Define a Named Range**

Do you want your charts or pivot tables to automatically expand as you add more data? Set up a Named Range that uses the OFFSET formula. As shown in Figure 4, a named range that refers to =OFFSET(Sales!\$A\$1,0,0,COUNTA(Sales!\$A:\$A),3) will grow to include new data added at the bottom of the data set.

While the OFFSET function can do many things, you'll find that it's usually most effective when using only one aspect of the function at a time. ■

*Bill Jelen is the host of MrExcel.com. IMA members can download a free PDF of his latest book at [www.MrExcel.com/ima.html](http://www.MrExcel.com/ima.html). Send questions for future articles to [IMA@MrExcel.com](mailto:IMA@MrExcel.com).*